プログラムのページ

担当 伊 理 正 夫

6408. 方程式の実根の計算に連分数展開を利用する方法（Lagrange）のプログラム

戸田 英雄（電気試験所応用数学研究室）

1. 方程式

$f(x)=a_0x^n+a_1x^{n-1}+\cdots\cdots+a_{n-1}x+a_n=0$ （1）

（ここで $a_j(j=0, 1, \cdots\cdots, n)$ は実数とする.）

2. 解法（Lagrangeによる；たとえば[1]参照）

方程式（1）の実根を α とし，その近似値がなんらかの方法により既知とする．計算の手順は次のとおり．

i） $k_0<\alpha<k_0+1$ なる整数 k_0 を定める.

ii） $\alpha=k_0+1/\alpha_1$ とおいて，α_1 を満す方程式

$f_1(x_1)\equiv x_1{}^n\cdot f(k_0+1/x_1)$ （2）

を求める（$\alpha_1>1$ である）.

iii） $k_1<\alpha_1<k_1+1$ なる整数 k_1 を定める.

iv） $\alpha_1=k_1+1/\alpha_2$ とおいて，α_2 を満す方程式

$f_2(x_2)=x_2{}^nf(k_1+1/x_2)=0$ （3）

を求める（$\alpha_2>1$）.

v） このような操作をつづけて，（1）の実根 α を次のように連分数に展開する.

$$\alpha\fallingdotseq k_0+\frac{1}{k_1+}\frac{1}{k_2+}\cdots\cdots+\frac{1}{k_{m-1}+}\frac{1}{k_m}. \quad (4)$$

vi） （4）の近似分数

$$\frac{p_{m+1}}{q_{m+1}}=\frac{p_mk_m+p_{m-1}}{q_mk_m+q_{m-1}} \quad (5)$$

は次の関係式から求められる.

$p_0=1,\ \ p_1=k_0.$ （6）

$p_j=p_{j-1}*k_{j-1}+k_{j-2}\ \ (j\geq 2).$ （7）

$q_j=q_{j-1}*k_{j-1}+k_{j-2}\ \ (j\geq 2).$ （8）

vii） α を連分数展開して，その近代分数を $\frac{p_{m+1}}{q_{m+1}}$ とすると $\left|\alpha-\frac{p_{m+1}}{q_{m+1}}\right|<\left(\frac{1}{q_{m+1}}\right)^2$ であるので，α の近似値の誤差が評価できる.

3. 連分数による整数計算のプログラム

以下は OKITAC 5090 のための ALGOLIP で書かれたプログラムである．（注 1）に内容のあらましを説明してある．（注 2）にデータの与え方について付記した.

1） 高木貞治，初等整数論講義．共立出版社.

```
begin integer I, N, X, FX, J, M, XUP, P2, Q2,
      KI, X0, IREV, WS;
  real ERROR, ROOT, EPS;
  integer array A, B, H[0:10], K[0:20];

  procedure POLY;
  begin for I:=1 step 1 until N do
                B[I]:=X*B[I-1]+H[I];
     FX:=B[N]
  end;

  procedure SYNDIV;
  begin CRLF(1); PRINTSTRING('SYNTHE');
    PRINTSTRING ('TIC DI');
    PRINTSTRING('VISION'); CRLF(2);
    for J:=1 step 1 until N do
    begin for I:=1 step 1 until M do
                  H[I]:=H[I-1]*X+H[I];
       M:=M-1
    end
  end;

  procedure UPPER;
  begin integer MAX;
    real SF, FH, QH; array FHH[0:10];
    MAX:=0; SF:=0.0; FH:=FLOAT(H[0]);
    for I:=0 step 1 until N do
      FHH[I]:=FLOAT(H[I])/FH;
    for I:=0 step 1 until N do
    begin QH:=FHH[I];
      if QH≧0 then SF:=SF+QH
      else begin XUP:=1+FIX(ABS(QH)/SF);
                 if MAX<XUP then MAX:=XUP
         end
    end;
    XUP:=MAX; CRLF(1);
    PRINTSTRING('  XUP=');
    PRINTINTEGER(XUP); CRLF(1);
```

```
end;

procedure FRACTION;
begin integer P0, P1, Q0, Q1; real FQ2;
  CRLF(1); PRINTSTRING('   FRCT');
  CRLF (1);
  P0:=1; P1:=K[0]; P2:=1;
  Q0:=0; Q2:=1; Q1:=1;
  for I:=2 step 1 until KI do
  begin P2:=P1*K[I-1]+P0;
        Q2:=Q1*K[I-1]+Q0;
        PRINTINTEGER(I);
        PRINTINTEGER (P2);
        PRINTINTEGER(Q2); CRLF(1);
        P0:=P1; P1:=P2; Q0:=Q1;
          Q1:=Q2
  end;
  FQ2:=FLOAT(Q2);
  ROOT:=FLOAT(P2)/FQ2;
  ERROR:=1.0/(FQ2*FQ2);
  PRINTSTRING(' ROOT ');
  PRINTREAL (ROOT); CRLF(1);
  PRINTSTRING(' ERROR');
  PRINTREAL (ERROR); CRLF(1)
end;

procedure REVERSE;
begin integer TEMP1, TEMP2;
  CRLF(1); PRINTSTRING('REV   ');
  CRLF(1);
  for I:=0 step 1 until IREV do
  begin TEMP1:=H[I];
        TEMP2:=H[N-I];
        H[I]:=TEMP2;
        H[N-I]:=TEMP1
  end
end;

procedure LAGRANGE;
begin integer X1, X2, F1, F2, F12, S;
  X:=X0; M:=N; KI:=0;
  for I:=0 step 1 until N do H[I]:=A[I];
REPT: CRLF(1); PRINTSTRING('   KI=');
  PRINTINTEGER(KI); PRINTINTEGER(X);
    CRLF(1);
    K[KI]:=X; M:=N; KI:=KI+1;
    SYNDIV; IREV:=(N-1)÷2; REVERSE;
    for I:=0 step 1 until N do
    begin PRINTINTEGER(H[I]); CRLF(1)
    end;
    UPPER;
    X1:=XUP; S:=1; B[0]:=H[0]; X:=X1;
    POLY;
    F1:=FX; PRINTINTEGER(X1);
    PRINTINTEGER(F1); CRLF(1);
    if F1=0 then
    begin ERROR:=0.0; go to TEST end;
  STEP: X2:=X1+S; X:=X2;
    POLY;
    F2:=FX; PRINTINTEGER(X2);
    PRINTINTEGER(F2); CRLF(1);
    if F2=0 then
    begin ERROR:=0.0; go to TEST end;
    if S>0 then X:=X1 else X:=X2;
    if ABS(F1)≧ABS(F2) then
             F12:=F1÷F2
             else F12:=F2÷F1;
    if F12<0 then go to CONTINUED;
    if ABS(F1)≧ABS(F2) then
      begin F1:=F2;
      X1:=X2; go to STEP
      end;
    if S>0 then
      begin S:=-S; X1:=X2; go to STEP
      end;
    go to COMP;
  CONTINUED: FRACTION;
  TEST: if ERROR>EPS then go to REPT;
  end;

PREP: CRLF(1); SPACE(2);
  PRINTSTRING('CONTIN');
  PRINTSTRING('UED FR');
  PRINTSTRING('ACTION'); SPACE(2);
  PRINTSTRING('1964/5');
  PRINTSTRING('H. TODA'); CRLF(2);
START: CRLF(3);
```

```
  N:=READINTEGER;
  PRINTSTRING('     N= ');
  PRINTINTEGER(N); CRLF(1);
  PRINTSTRING('A(  )= '); CRLF(1);
  EPS:=₁₀-9;
  for I:=0 step 1 until N do
  begin WS:=A[I]:=READINTEGER;
        PRINTINTEGER(WS); CRLF(1)
  end; CRLF(1);
  X0:=READINTEGER;
CALL: LAGRANGE;
  CRLF(1)
  PRINTSTRING('ROOT = ')
  PRINTINTEGER(P2);
  PRINTINTEGER(Q2); CRLF(1);
  PRINTREAL(ROOT); CRLF(1);
  PRINTSTRING(' ERROR= ');
  PRINTREAL(ERROR);
  go to PREP;
COMP: PRINTSTRING('COMPLE'); CRLF(1);
  go to PREP;
end
```

データの例（四例）

3, 1, 0, −7, 7, 1,
3, 1, 0, −7, 7, −4,
3, 1, 0, −2, −5, 2,
4, 1000000, −4860000, 8857100, −7173846,
2178871, 1,

(注 1) プログラムのあらまし.
procedure POLY: 多項式 $f(x)$, $f_1(x)$, $f_2(x)$ の計算
procedure SYNDIV; 組立除法によい $f_1(x)$, $f_2(x)$,…の計算.
procedure UPPER: $f_1(x)=0$, $f_2(x)=0$,…の正根の上界の計算.
procedure LAGRANGE: はさみうち法により k_1, k_2,……の決定.
COMP: 実根のないときはここにとぶ.
(注 2) プログラムの最後に四組のデータがある. たとえば一番はじめのは例1のもので, 3（次数）, 1, 0, −7, 7（$x^3+0\cdot x^2-7x+7=0$ の係数）, 1（$k_0=1$ を与える）の順にテープから与える. 以下同様.

4. 数値例 3. で示したプログラムにその最後に付記したデータを与えて計算した結果は次のとおりである.

例 1.[1] $x^3-7x+7=0$.

この方程式は区間 (1, 2) に2個の正根と, 区間 (−4, −3) に1個の負根がある.

$k_0=1$ の場合（第1のデータによる）:

$$\alpha=1+\frac{1}{2+}\frac{1}{1+}\frac{1}{4+}\frac{1}{20+}\frac{1}{2+}\frac{1}{3+}\frac{1}{1+}\frac{1}{6+}\frac{1}{10}$$
$$=242902/179013=\cdot1356895867_{10}1.$$

誤差$=\cdot3120547937_{10}-10$.

$k_0=-4$ の場合（第2のデータによる）:

$$\beta=-4+\frac{1}{1+}\frac{1}{19+}\frac{1}{2+}\frac{1}{3+}\frac{1}{1+}\frac{1}{6+}\frac{1}{10+}\frac{1}{5}$$
$$=-196707/64517=-\cdot3048917339_{10}1.$$

誤差$=\cdot2402435133_{10}-9$.

例2.[2] $x^3-2x-5=0$

この方程式は区間 (2, 3) に一個の実根がある.

$k_0=2$ の場合（第3のデータによる）:

$$\alpha=2+\frac{1}{10+}\frac{1}{1+}\frac{1}{1+}\frac{1}{2+}\frac{1}{1+}\frac{1}{3+}\frac{1}{1+}\frac{1}{1+}\frac{1}{12+}\frac{1}{3+}\frac{1}{5}$$
$$=269175/128512=\cdot2094551481_{10}1.$$

誤差$=\cdot6054978915_{10}-10$.

例 3.[3]

$$1000000x^4-4860000x^3+8857100x^2-7173846x+2178871=0.$$

この方程式で, もし常数項が (2178871+1) のときは 1.20, 1.21, 1.22, 1.23 という近接した4個の実根をもつことが知られている. (1, 2) で1に一番近い根 α を求める（第4のデータによる).

$$\alpha=1+\frac{1}{5+}\frac{1}{4+}\frac{1}{5+}\frac{1}{1+}\frac{1}{5+}\frac{1}{4+}\frac{1}{9+}\frac{1}{4}$$
$$=144244/121127=\cdot1190849273_{10}1.$$

誤差$=\cdot6815819461_{10}-10$.

（昭和 39 年 9 月 28 日受付）

1) Lagange の与えた例.
2) Newton の与えた例.
3) 岡本の与えた例.